
Part 1: Why is my model running so slow!?! The Profiler extension
I think it’s fair to say that all modelers want their models to run faster. A faster run time means we can do more tests during development, run the model longer, run a broader suite of variables during sensitivity analyses, and get results from the final complete set of runs with Netlogo’s BehaviorSpace faster. My modeling mentor always told us that archaeological models were “over modeled and under run” and so making your model run faster is a good way to run it more and make sure you understand what it is doing.

However, our models are limited by the processing speed of the computer. This is affected by the processor itself, but also the RAM and the hard disk speed depending on what you’re asking the model to do. For the sake of this tutorial we’ll keep it simple and just think about this as the total time it takes to accomplish the tasks we set for it. A simple count of all the things that are asked of the model gives a good estimate. Most of the tasks and little math operations can be counted as roughly equivalent (for exceptions see the end of the post). 

Before we get to Netlogo’s Profiler extension, think about how many patches and how many agents there are in your model. Probably you have more patches than agents. So any time you ask all the patches to do something, it’s probably going to take longer to process than asking all the agents to do something. For this reason I tend to try to get most things done by the agents themselves, even if it is updating a patch variable.

For example, this:
ask patches [if count agents-here > 1 [set resources resources - ( count agents-here * harvest-rate )]]

will take much longer than this:
ask agents [ask patch-here [set resources resources - harvest-rate]] 

even though both are accomplishing the same task. 

Let’s code in a simple foraging model to get us started:
breed [agents agent]
globals [harvest-rate]
agents-own [backpack]
patches-own [resources]

to setup
  ca
  reset-ticks
  set harvest-rate 10
  ask patches [
    set resources random 100
    set pcolor scale-color green resources min [resources] of patches max [resources] of patches
    ]
  create-agents 10
  [
    move-to one-of patches
    set color red
  ]
end
to go
  ask agents [
    ;;;;;;;;;;;;;;;;;;;;;;;;;;;;;; consume resources
    set backpack backpack + harvest-rate
    ask patch-here [set resources resources - harvest-rate ]
    ;;;;;;;;;;;;;;;;;;;;;;;;;;;;;; move to a new cell 
    let p one-of neighbors
    if [resources] of p > [resources] of patch-here [move-to p]
  ]
  ;;;;;;;;;;;;;;;;;;;;; Update the display to reflect the change in resources
  ask patches [
    set pcolor scale-color green resources min [resources] of patches max [resources] of patches
  ]
  tick
end

To quantify the model’s processing speed, we need to find out which parts of the model are asked to be performed and how many times. This is where Netlogo’s Profiler extension comes in. To enable it, add the following to the very top of your model’s code:
extensions [profiler] 

Next add a button to the interface of the model, call it Profiler, and then in the code box enter (copied directly from Netlogo’s programming guide):

setup                  ;; set up the model
profiler:start         ;; start profiling
repeat 30 [ go ]       ;; run something you want to measure
profiler:stop          ;; stop profiling
print profiler:report  ;; view the results
profiler:reset         ;; clear the data

This is assuming you have set up your favorite model in the standard way with the initialization procedure called setup, and the iterating part of the model called go. Now when we push the button it will run the model through 30 ticks of the model and then drop the processing time results into the Command Center of Netlogo.

BEGIN PROFILING DUMP
Sorted by Exclusive Time
Name                               Calls Incl T(ms) Excl T(ms) Excl/calls
GO                                     30   2339.607   2339.607     77.987

Sorted by Inclusive Time
GO                                     30   2339.607   2339.607     77.987

Sorted by Number of Calls
GO                                     30   2339.607   2339.607     77.987
END PROFILING DUMP

The results are structured in three tables which are really just one table re-sorted by each of the table columns. Each row represents a model procedure that was called at least once. The first column is the number of times that procedure was “called”. The second is the “Inclusive time” in milliseconds for that procedure, or the total time the model spent within that procedure AND the procedures it called (e.g. the inclusive time for “go” is the total run time). The second column is the “Exclusive time”, which is the total time spent just on that procedure WITHOUT any procedures it called. The last is the exclusive time per call (rather than summed over all the calls). I find the Exclusive Time column the most useful in terms of finding out why my model is taking so long. 

Now, if you have designed your model such that everything happens within one big go procedure, you’re not going to get very interesting results here. To make the most of Profiler, you’re going to have to break up the model into separate procedures. This is good practice for coding anyway since it helps to be able to see the overall structure of the model in go (almost like a table of contents), and have the different parts of the model divided up into different segments of code.
 
As an example of how to do this, I have re-programmed the go code from above (for posterity, rename your old go procedure instead of deleting it):
to go
ask agents [
    ;;;;;;;;;;;;;;;;;;;;;;;;;;;;;; consume resources
    consume          ;this is our new consume procedure, which is defined below
    ;;;;;;;;;;;;;;;;;;;;;;;;;;;;;; move to a new cell 
    move             ;our new move procedure
  ]
  update-patches     ;and the update-patches procedure
  tick
end

to consume
  set backpack backpack + harvest-rate
  ask patch-here [
    set resources resources - harvest-rate 
]
end

to move 
  let p one-of neighbors 
  if [resources] of p > [resources] of patch-here [move-to p]
end

to update-patches
  ask patches [
    set pcolor scale-color green resources min [resources] of patches max [resources] of patches
  ]
end
 
Now when we click our Profiler button it will divide the processing time into the time spent on go, move, consume, and update-patches. This is useful since now we can see which is taking the most processing time and we can re-code our model to make it faster. 

BEGIN PROFILING DUMP
Sorted by Exclusive Time
Name                               Calls Incl T(ms) Excl T(ms) Excl/calls
UPDATE-PATCHES                        30   2221.975   2221.975     74.066
CONSUME                              300      1.804      1.804      0.006
GO-AFTER                              30   2226.679      1.603      0.053
MOVE                                 300      1.297      1.297      0.004

You can easily see from this table that update-patches is the part taking the most time even though it is being called much less often than consume and move. So why is this? Well if you look at the code you can see that within the procedure I’ve asked every patch to look at every patch twice (once for min, once for max)! This is a ridiculously computationally intensive way to do something as simple as changing the patch colour. A better way is like this:

to update-patches-faster
  let min-val min [resources] of patches
  let max-val max [resources] of patches
  ask patches [
    set pcolor scale-color green resources min-val max-val
  ]
end 

On my machine this took 2000 ms the first way and only 12 ms the second way. An even faster method would be to have your agents update the pcolor of the patch they’ve just harvested (bonus points to the first person to figure out how to do this and tweet the solution to @cdwren). This has the advantage of most patches not being asked to do anything for most ticks, and as expected it reduced the processing time to 0.4 ms.

If you’re unsure which of your different coding methods is going to run faster, then make a new profiler button to check it out! On the interface, make a new button, call it profile-modules, and do the same as before but with go replaced by the name of your modules:
setup                  ;; set up the model
profiler:start         ;; start profiling
repeat 30 [ update-patches update-patches-faster ]       ;; run something you want to measure
profiler:stop          ;; stop profiling
print profiler:report  ;; view the results
profiler:reset         ;; clear the data
Now when you click the button, it will give quickly you a specific break-down of those options without the rest of your model getting in the way. 

Lastly, my top 6 list of computational time sucks to avoid in NetLogo:
1) ask patches
Get the agents to update patches whenever possible. Even if the code to do it is longer, it’ll go faster.
2) Interface plots 
Computational time for plots are included to the go section of the profiler report. While necessary, plots take a lot of time. Add a plots-on? switch to the interface, and an if plots-on? to the plot code and turn them off when they’re unneeded.
3) Useful for debugging, but make sure to comment out those show, type and print lines as writing to Command Center slows things down.
4) Think simpler. Writing a simple model is hard, but reducing the complexity of agent behavior will speed everything up. Keep only what’s necessary and sufficient.
5) Mean, min, max [variable] of patches
Already covered above, but use temporary variables to calculate these once instead of having every patch or agent repeat it.
6) let mypatchset patch-set patches with [pcolor = red]
ask mypatchset in-radius 3 [set pcolor = blue]
This is a weird one that has to do the way NetLogo is programmed, but find another way. This will make your model crawl.

Download my sample model here: https://www.openabm.org/model/4557/version/1. 

Part 2: What is my model doing?!!?!   Using BehaviorSpace
What and why BehaviorSpace
Eventually you will have a functioning model with at least a handful of variables. At this point you have most likely mucked about with the variable settings sliding your sliders up and down and running and re-running your model. That experimentation is all well and good, but is not so great for quantitative analysis or even a true understanding of your model’s dynamics. For this we need to be more systematic. This is where BehaviorSpace comes in. 

BehaviorSpace is important enough to NetLogo, that it should really be on a fourth interface tab right after “Code”. Put simply, BehaviorSpace experiments will run your model over and over again with different variable settings, and will then export the results into a giant table. It has a few important purposes that I will briefly describe, then I will describe how to go about creating your experiments, and what the results will look like. 

Sensitivity analysis
The purpose of this is to explore how sensitive your model is to different variables. If you increase variable A by some amount, what is the effect on some output measure? Does it completely change the result, or does it just increase the output a little? What happens if you increase A & B? The sensitivity analysis repeats model runs with different combinations of variables so that:
a) you can make sure the model does what you think it does
b) you’ll notice if an odd variable combination produces some interesting and unexpected emergent phenomenon, and 
c) you can choose reasonable variable values and ranges for testing your hypothesis

Testing a hypothesis
Eventually you’ll have a variable or two that will allow you to address your research question. At this point you’ll want to hold most of your variables constant (at reasonable levels determined by your sensitivity analysis), and just vary the ones you’re interested in.

Producing a quantitative output
Sure you could have an output measure pop into the command center at the end of a run but that’s not a very practical way of collecting data. BehaviorSpace experiments will export a table with your variable settings and output measures that you can then open in R, Excel, or your data analysis package of choice.

Creating and running an experiment
Preparation
Before you open BehaviorSpace, make sure the variables you’re interested in are controlled in the Interface tab and not just numbers in the Code tab (i.e. turn them into sliders, choosers, switches, etc.). This is required so that BehaviorSpace can see your variables. For example, from the Profiler model you finished above, you could turn number-agents, harvest-rate, and max-resources into sliders. You’ll also need to change the relevant lines of code:
globals []
…
  ;set harvest-rate 10
  ask patches [
    set resources random max-resources
    set pcolor scale-color green resources min [resources] of patches max [resources] of patches
  ]
  create-agents nragents
  [
    move-to one-of patches 
    set color red
  ]
... just for fun, let’s put in a move-type Chooser on the interface with two move options, “random” and “maximizer” as well: 
;move
ifelse move-type = "random" [move-random][move-maximizer]
...
to move-random 
  let p one-of neighbors
  if [resources] of p > [resources] of patch-here [
    move-to p
    set backpack backpack - 1
    ]
end

to move-maximizer
  let p max-one-of patches in-radius 2 with [count agents-here = 0] [resources]
  move-to p
  set backpack backpack - 1
end       
To complete this circle of resource harvest and regrowth, I also added code to regrow the resource value of patches, and modified the consume procedure to not consume resources if the patch has too few. I’ll let you figure those bits out. 

You’ll also want to decide on your output measures. BehaviorSpace assumes that you’ll want to count turtles, but the rest is up to you. Think hard about the simplest possible way to assess the result of a model run, hopefully you’ll get it down to one or two measures like:
mean [resources] of patches
mean [backpack] of agents
Also think about whether you need these measures at every tick of the model, or just at the end (hint: choose just at the end if you can get away with it).

Create the experiment
Open BehaviorSpace (Tools menu) and click New.
[image: ]If you followed the Preparation instructions, then a lot of the experiment will be created for you. This part is pretty clearly explained in the window including some fine print instructions on different ways to “Vary variables” as a series of specific values or as incrementing ranges [start increment end]. Remember to put quotes around text strings like variable names. 

If you put: [“nragents” 10 20 30] you’ll have 3 runs. If you then add: [“harvest-rate” [0.1 0.1 0.3]], you’ll now have 9 runs.

Put your output measures from above in the “Measures” box. If you want to repeat variable configurations then put how many in the “Repetitions” box. If at all possible, uncheck the “Measure runs at every step” box. This will make analysis much simpler.

If you don’t want your model to run forever, you will need to either put in a time limit (in ticks) or a Stop condition structured like a reporter, count agents = 0, or both.

The Final commands box can be useful for running export-view to take a snapshot of the map or export-plot to record a variable’s change during a run, perhaps instead of measuring something at every tick. In these cases you’ll need to specify filenames using (word “agents_” nragents “_” behaviorspace-run-number “.png”) so that you don’t overwrite your file with every run. behaviorspace-run-number is a unique number for each run, so this could help unless you’re running multiple experiments. date-and-time is another useful variable that would solve that problem (but will make long filenames).

Click ok and you’re done. The experiment name should show up in the box with the number in brackets representing how many runs of the model your variable set requires. If the number seems too big, go back and reduce the range of a variable, or maybe just shorten the length of the runs. 

Running the experiment
Select your experiment and click Run. There are two output options, I find Table to be more useful personally. As a speed bonus, if you have a multiple multi-core processors you can choose to run multiple experiments at once. If you want to keep using your computer while BehaviorSpace is running, subtract one from the number given or it might be un-usably slow. 
[image: ]
Click ok, then chose a location to save your output table. Your experiment should now be running. If you want it to go even faster turn off the view, plots and monitors. Now sit back and wait. Depending on the model and the number of runs this could be an hour, or a couple weeks. 

I don’t really recommend it due to write permissions, but you can open the output table before your experiment is finished to see that it’s recording everything correctly.
Analysing the output table
The output is a csv (comma separated values) file which can be opened in notepad or Excel, or imported into R. The top few lines give some meta-data about the run. Below that, each row is a run, each column is either an input variable or an output measure. Try to figure out which parameter settings result in your agents not over-exploiting their environment and starving to death. Is it better for them to random or maximizers? You’re on your own for this part. Happy plotting!

